

Exercises for SAS Course – AEM 2011

Assignment 1)

In order to get experience with the different matrix operating procedures, do the following:

- Create a 3x3 matrix (name: seven) with all 7's.
- Create a column vector (name: vector) with the numbers from 5 to 15
- Create a matrix (name: three) with the following values

$$\begin{bmatrix} 5 & 7 & 1 \\ 2 & 7 & 8 \\ 20 & 21 & 4 \end{bmatrix}$$
- Find the inverse matrix to three (name: Inv_three).
- Multiply seven with three. (name: sevenxthree) (Try to multiply three with column and see what the log says)
- Multiply seven and three elementwise. (name: sevenxthree)
- Pick out the 2nd row of three and call it row. Then pick out the 3rd column of seven and call it column. Multiply row and column (name: rowxcolumn).
- Sum across the columns of three. (name: sum_col_three)

Assignment 2)

In the workbook stocks.xls you will find the log-returns on 3 stocks and 1 index; OMX C20, Carlsberg, SAS and TDC. There are 261 returns.

Import the data into a SAS and create a regression module calculating beta for each of the three stocks.

The results should be: Carlsberg: 0,6979644; SAS: 1,3397; TDC: 0,1012995

Assignment 3)

Create a module to estimate the market model (both the betas and the constant) for the three stocks from assignment 2. This time do it using a do-loop, so you easy can alter the code to calculate the market model for 100 stocks.

The results should be:

	Constant:	Beta:
Carlsberg:	0,000216	0,6985586
SAS:	-0,003118	1,3311279
TDC:	0,0010492	0,1041852

Solutions:

```
/*Assignment 1*/
proc iml;
seven= j(3,3,7);
vector=t(5:15);
three={5 7 1, 2 7 8, 20 21 4};
Inv_three=inv(three);
sevenxthree=seven*three;
sevenxxthree=seven#three;
row=three[2,];
column=seven[,3];
rowxcolumn=row*column;
sum_col_three=three[,+];
print seven;
print column;
print three;
print Inv_three;
print sevenxthree;
print sevenxxthree;
print row;
print column;
print rowxcolumn;
print sum_col_three;
quit;

/*Assignment 2*/
proc import out=stocks
datafile= '\\okf-filesrv1\exemp\SAS\Empirical Finance\stocks.xls' replace;
sheet='sheet1';
getnames=yes;
mixed=no;
run;
proc print data=stocks;
run;

proc iml;
use work.stocks;
read all var {Carlsberg} into Carlsberg;
read all var {SAS} into SAS;
read all var {TDC} into TDC;
read all var {C20} into C20;
print carlsberg;
Beta_Carlsberg=inv(C20`*C20)*C20`*Carlsberg;
Beta_SAS=inv(C20`*C20)*C20`*SAS;
Beta_TDC=inv(C20`*C20)*C20`*TDC;
print Beta_Carlsberg;
print Beta_SAS;
print Beta_TDC;
quit;

/*Assignment 3*/
PROC IML;
USE stocks;
READ all var _num_ into return;
print return;
```

```
N=nrow(return);
beta=J(3,2,0);
X=J(N,1,1)||return[,1];
do i=1 to 3 by 1;
    y=return[,1+i];
    beta[i,]=(INV(X`*X)*X`*y)`;
end;
print beta;

create regdata from beta[colname={"alpha" "beta"}];
append from beta;

quit;
```